Dubin 3.2.17 Melting Fat 11-15-16

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[69]:=
```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

The original version of this notebook is *Dubin 4.2.2 Grilling Refined 2 (6-18-2005)*

Purpose

I solve a heat equation problem from Chapter 3 of *Numerical and Analytical Methods for Scientists and Engineers, Using Mathematica*, Daniel Dubin. The specific Problem 3.1(17) considers the melting of a slab of fat in hot water.

Part (a) Solution of PDE and determination of time for temperature to rise to specified level

I construct an Association that encapsulates information about this problem, and then apply the function *DSolveHeatEquation* that attempts to solve this problem directly using the Mathematica function *DSolve*.

In[71]:=

```
A1 =
  Module[{description, pde, bcL, bcR, ic, eqns, depVar,
    assumptions, substitutions, simplifications, names, values},
   description = "Dubin problem 3.1(17) Melting Fat\nHomogeneous heat
      equation, inhomogeneous Dirichlet boundary conditions";
   pde = D[T[x, t], t] - χ D[T[x, t], {x, 2}] == 0;
   bcL = T[0, t] == Tw; (* inhomogeneous Dirichlet *)
   bcR  = T[L, t] == Tw; (* inhomogeneous Dirichlet *)
   ic =  T[x, 0] == T0;
   eqns = {pde, bcL, bcR, ic};
   depVar = T[x, t];
   assumptions = {L > 0, χ > 0};
   substitutions = {K[1] → n};
   simplifications = {n ∈ Integers};
   values = {description, pde, bcL, bcR, ic,
     eqns, depVar, assumptions, substitutions, simplifications};
   names = {"description", "pde", "bcL", "bcR", "ic", "eqns",
     "depVar", "assumptions", "substitutions", "simplifications"};
   AssociationThread[names, values]];

Module[{soln, G},
 soln = DSolveHeatEquation[A1];
 AppendTo[A1, "soln" → soln];
 Print@ShowPDESetup[A1];
 A1["soln"]]
```



Out[72]=

$$T[x, t] \to Tw + \sum_{n=1}^{\infty} -\frac{2\left(-1+(-1)^n\right) e^{-\frac{n^2 \pi^2 t \chi}{L^2}} (T0 - Tw)\, Sin\left[\frac{n \pi x}{L}\right]}{n \pi}$$

DSolve immediately solves this problem

In[73]:=

```
Clear[TSoln];
TSoln[x_, t_, χ_, L_, T0_, Tw_, nMax_] :=
```

$$Tw + \sum_{n=1}^{nMax} -\frac{2\left(-1+(-1)^n\right) e^{-\frac{n^2 \pi^2 t \chi}{L^2}} (T0 - Tw)\, Sin\left[\frac{n \pi x}{L}\right]}{n \pi} \text{ // Activate}$$

In[75]:=

```
Module[{χ = 10^-6 (* m^2/s *), L = 0.02 (* 0.02 m *),
  T0 = 5 (* deg C *), Tw = 60 (* deg C *), nMax = 20},
 Plot3D[TSoln[x, t, χ, L, T0, Tw, nMax], {x, 0, L}, {t, 0, 100},
  ColorFunction → "TemperatureMap", AxesLabel → {Stl["x"], Stl["t"], Stl["T"]},
  Mesh → False, Boxed → False, PlotLegends → Automatic]]
```

Out[75]=



In[76]:=

```
Module[{χ = 10^-6 (* m^2/s *), L = 0.02 (* 0.02 m *), T0 = 5 (* deg C *), Tw = 60
   (* deg C *), Ttarget = 55 (* deg C *), nMax = 50, tDone, lab, refLine},
  refLine = {Directive[Black, Dashed], Line[{{0, Ttarget}, {120, Ttarget}}] };
  tDone = FindRoot[TSoln[0.01, t, χ, L, T0, Tw, nMax] == Ttarget, {t, 100}][[1, 2]];
  lab =
   Stl@StringForm["Internal temperature as function of time\ntarget temperature ``
       degrees reached after `` sec", Round@Ttarget, NF2@tDone];
  Plot[{TSoln[0.01, t, χ, L, T0, Tw, nMax]}, {t, 0, 120},
   ColorFunction → "TemperatureMap", AxesLabel → {Stl["t"], Stl["T"]},
   Epilog → refLine, PlotLabel → lab]]
```

Out[76]=



Part (b) Analytical approximation of heating time

Consider only the first term in the series expansion for T(x, t)

In[77]:=

```
wb[1] = Tapprox[x, t] == TSoln[x, t, χ, L, T0, Tw, 1]
```

Out[77]=

$$T_{approx}[x, t] == Tw + \frac{4\, e^{-\frac{\pi^2 t \chi}{L^2}} (T0 - Tw)\, \mathrm{Sin}\left[\frac{\pi x}{L}\right]}{\pi}$$

At the midpoint

In[78]:=

```
wb[2] = wb[1] /. x → L/2
```

Out[78]=

$$T_{approx}\left[\frac{L}{2}, t\right] == \frac{4\, e^{-\frac{\pi^2 t \chi}{L^2}} (T0 - Tw)}{\pi} + Tw$$

Let t_{done} = the time the target temperature is reached.

In[79]:=
```
wb[3] = wb[2] /. t → tDone /. Tapprox[L/2, tDone] → Ttarget
```

Out[79]=

$$\text{Ttarget} == \frac{4\, e^{-\frac{\pi^2\, \text{tDone}\, \chi}{L^2}} (\text{T0} - \text{Tw})}{\pi} + \text{Tw}$$

In[80]:=
```
wb[4] = Solve[wb[3], tDone][[1, 1]]
```

Out[80]=

$$\text{tDone} \to \text{ConditionalExpression}\left[\frac{1}{\pi^2 \chi} L^2 \left(2\, i\, \pi\, C[1] + \text{Log}\left[-\frac{4\, (\text{T0} - \text{Tw})}{\pi\, (-\text{Ttarget} + \text{Tw})}\right]\right),\ C[1] \in \text{Integers}\right]$$

Choose the principal branch of Log

In[81]:=
```
wb[4] = wb[4] /. C[1] → 0
```

Out[81]=

$$\text{tDone} \to \frac{L^2\, \text{Log}\left[-\frac{4\, (\text{T0}-\text{Tw})}{\pi\, (-\text{Ttarget}+\text{Tw})}\right]}{\pi^2 \chi}$$

I check this against the PDE solution

In[82]:=
```
Module[{χ = 10^-6 (* m^2/s *), L = 0.02 (* 0.02 m *),
  T0 = 5 (* deg C *), Tw = 60 (* deg C *), Ttarget = 55
  (* deg C *), tDoneAnalytic, tDoneExact = 106.95 , info},
 tDoneAnalytic = (L^2 Log[-(4 (T0-Tw))/(π (-Ttarget+Tw))])/(π^2 χ);
 info = {{tDoneExact, tDoneAnalytic, 100 (tDoneAnalytic - tDoneExact) / tDoneExact}};
 PrependTo[info, {"PDE solution", "n = 1 analytic approximation", "%diff"}];
 LGrid[info, "Comparing exact and approximate solutions"]]
```

Out[82]=

Comparing exact and approximate solutions

PDE solution	n = 1 analytic approximation	%diff
106.95	106.973	0.0217619

The n = 1 approximation is remarkably accurate.

Functions

In[66]:=

```
Clear[ShowPDESetup];
ShowPDESetup[A_] := Module[{top = 1.0, right = 1.0,
    boundaries, labels, textInterior, textIC, textBCL, textBCR},
   boundaries = Line /@ {{{0, 0}, {right, 0}},
      {{0, 0}, {0, top}}, {{right, 0}, {right, top}}};
   labels = Text[PhysicsForm[A[#[[1]]] ], #[[2]]] & /@
     {{"pde", {right / 2, top / 2}}, {"ic", {right / 2, 0.0}},
      {"bcL", {0.0, top / 2}}, {"bcR", {right, top / 2}}};
   Graphics[{Directive[Black, Thick], boundaries, labels}, Axes → False,
    AspectRatio → 0.25, ImageSize → 500, PlotLabel → Stl[A["description"]]]]
```

In[67]:=

```
Clear[DSolveHeatEquation];
DSolveHeatEquation[A_] :=
 Module[{soln},
  soln =
   DSolve[A["eqns"], A["depVar"], {x, t}, Assumptions → A["assumptions"]][[1, 1]];
  soln = soln //. A["substitutions"];
  soln = Simplify[soln, A["simplifications"]];
  soln]
```